Bipendulum

We study a bipendulum, namely a system composed of a bar where two pendula are fixed, one of length l_1 and one of length l_2. See J.-F. Pommaret, Partial Differential Control Theory, Kluwer, 2001, p. 569.

The appropriate Ore algebra for this example is the Weyl algebra A, where D_t is the differential operator w.r.t. time t:

```
A = OreAlgebra[t, Der[t], g, l1, l2]
```

$\mathbb{K}[t, g, l_1, l_2][D_t; 1, D_t]$

Next we enter the system matrix:

```
MatrixForm[R = ToOrePolynomial[{{l1 Der[t]^2 + g, 0, -g}, {0, l2 Der[t]^2 + g, -g}}, A]]
```

$$\begin{pmatrix} D_t^2 l_1 + g & 0 & -g \\ 0 & D_t^2 l_2 + g & -g \end{pmatrix}$$

In terms of equations, the bipendulum is defined by:

```
ColumnForm@Thread[ApplyMatrix[R, {x1[t], x2[t], u[t]}] == 0]
```

$-g\, u[t] + g\, x_1[t] + l_1\, x_1''[t] == 0$
$-g\, u[t] + g\, x_2[t] + l_2\, x_2''[t] == 0$

We compute the formal adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} D_t^2 l_1 + g & 0 \\ 0 & D_t^2 l_2 + g \\ -g & -g \end{pmatrix}$$

Let us check whether or not the left A-module M is torsion-free, i.e. whether or not the system admits some autonomous elements.

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{ \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & D_t^2 l_2 + g & -g \\ -D_t^2 l_1 - g & 0 & g \end{pmatrix}, \begin{pmatrix} D_t^2 g\, l_2 + g^2 \\ D_t^2 g\, l_1 + g^2 \\ D_t^4 l_1 l_2 + D_t^2 g\, l_1 + D_t^2 g\, l_2 + g^2 \end{pmatrix} \right\}$$

Since the first matrix is identity, M is torsion-free. Let us now compute the obstructions for M to be projective:

```
P = ObstructionToProjectiveness[R, A]
```

$\{\{\{-g^2 l_1 + g^2 l_2\}, \{D_t^2 g\, l_2 + g^2\}, \{D_t^2 g\, l_1 + g^2\}, \{D_t^4 l_1 l_2 - g^2\}\}\}$

Let us deduce the obstructions in the parameters g, l_1, l_2 for M to be a projective left A-module.

```
OreIntersection[Flatten@P, {g, l1, l2}, A]
```

$\{-g^2 l_1 + g^2 l_2\}$

We obtain that M is a projective left A-module, i.e. the system is flat, if and only if $l_1 - l_2$ is not equal to zero.

If $l_1 - l_2 \neq 0$, let us compute flat output.

```
B = OreAlgebra[t, Der[t]]
```

$\mathbb{K}[t][D_t; 1, D_t]$

```
T = LeftInverse[Q, B]
```

$$\left\{\left\{\frac{l_1}{g^2\,(l_1 - l_2)}, -\frac{l_2}{g^2\,(l_1 - l_2)}, 0\right\}\right\}$$

A flat output ξ is then defined by:

```
ApplyMatrix[T, {x1[t], x2[t], u[t]}][[1]]
```

$$\frac{l_1\,x_1[t] - l_2\,x_2[t]}{g^2\,(l_1 - l_2)}$$

Finally a parametrization of the flat system is defined by

```
Thread[{x1[t], x2[t], u[t]} -> ApplyMatrix[Q, {ξ[t]}]]
```

$$\{x_1[t] \to g\,(g\,\xi[t] + l_2\,\xi''[t]),\ x_2[t] \to g\,(g\,\xi[t] + l_1\,\xi''[t]),$$
$$u[t] \to g^2\,\xi[t] + g\,l_2\,\xi''[t] + l_1\,(g\,\xi''[t] + l_2\,\xi^{(4)}[t])\}$$

If $l_1 - l_2 = 0$, then let us compute the autonomous elements of the new system. To do that let us introduce a new Ore algebra B, defined by

```
B = OreAlgebra[t, Der[t], CoefficientNormal → (Expand[# /. l1 → l2] &)]
```

$\mathbb{K}[t][D_t; 1, D_t]$

the matrix S of differential operators given by

```
MatrixForm[S = ToOrePolynomial[{{l1 Der[t]^2 + g, 0, -g}, {0, l2 Der[t]^2 + g, -g}}, B]]
```

$$\begin{pmatrix} l_2\,D_t^2 + g & 0 & -g \\ 0 & l_2\,D_t^2 + g & -g \end{pmatrix}$$

and the left B-module N finitely presented by S. Let us compute the torsion elements of N, i.e. the autnomous elements of the corresponding system.

```
MatrixForm[Sadj = Involution[S, B]]
```

$$\begin{pmatrix} l_2\,D_t^2 + g & 0 \\ 0 & l_2\,D_t^2 + g \\ -g & -g \end{pmatrix}$$

```
MatrixForm /@ ({Ann, Sp, Q} = Exti[Sadj, B, 1])
```

$$\left\{\begin{pmatrix} l_2\,D_t^2 + g & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & -1 & 0 \\ 0 & -l_2\,D_t^2 - g & g \end{pmatrix}, \begin{pmatrix} -g \\ -g \\ -l_2\,D_t^2 - g \end{pmatrix}\right\}$$

Since the first matrix is not identity, the system is not controllable (N is not torsion free) and we have one autonomous element, defined by the first row of Sp, namely θ, defined by

```
ApplyMatrix[{Sp[[1]]}, {x1[t], x2[t], u[t]}]
```

$\{x_1[t] - x_2[t]\}$

Moreover, θ satisfies the equation

```
Thread[ApplyMatrix[{{Ann[[1, 1]]}}, {θ[t]}] == 0]
```

$\{g\,\theta[t] + l_2\,\theta''[t] == 0\}$

Finally, this result can be obtained directly by means of the command AutonomousElements

```
AutonomousElements[S, {x1[t], x2[t], u[t]}, θ, B]
```

$\{\{\theta[1][t] \to x_1[t] - x_2[t],\ \theta[2][t] \to g\,u[t] - g\,x_2[t] - l_2\,x_2''[t]\},$
$\{g\,\theta[1][t] + l_2\,\theta[1]''[t] == 0,\ \theta[2][t] == 0\}\}$

The controllable part of the system is defined by:

```
Thread[ApplyMatrix[Sp, {x1[t], x2[t], u[t]}] == 0]
```

$\{x_1[t] - x_2[t] == 0,\ g\,u[t] - g\,x_2[t] - l_2\,x_2''[t] == 0\}$

Finally let us compute a flat output of the controllable part:

```
ApplyMatrix[LeftInverse[Q, B], {x1[t], x2[t], u[t]}][[1]]
```

$-\frac{x_2[t]}{g}$